<%
'--------------------------------------------------------------------------------------------
'Questa funzione serve a controllare che i file singoli non siano
'richiamati dal browser ma solo da OpenAsp, nel caso in cui il browser tenta di aprire tali file
'l'utente viene subito rimandato alla homepage del sito
'splitto la stringa SQL per recuperare la pagina che si cerca di aprire
i = split(request.ServerVariables("HTTP_URL"), "/")
'confronto il nome del modulo con la funzione Ubound che mi restituisce il massimo
'valore dell'array, quindi il nome della pagina richiesta dall'utente, se coincidono
'rimando alla index
if strComp("admin_locked.asp", i(Ubound(i)), 1) = 0 then
  response.redirect request.servervariables("HTTP_REFERER")
end if
'--------------------------------------------------------------------------------------------
%>
<div id="position">
      &raquo; <a href="admin.asp" class="testo">Control Center</a> &raquo; <%=traducisys("ling_occ_178")%>
</div>
<div id="page">
<%
'---------LIMITO IL BLOCCO SOLO AL SUPERADMIN----------
Call onlySADM()
'------------------------------------------------------
  Select case request.QueryString("pass")
  CASE 0, "":
  	IF SYSTEM_STATUS = 1 THEN
	%>
	<p style="text-align:center">
	<%=traducisys("ling_occ_194")%><br/><br/>
	<a title="<%=traducisys("ling_occ_195")%>" href="admin.asp?op=admin_locked&pass=2" class="testo"><%=traducisys("ling_occ_195")%></a>
	</p>
	<%
	else
	%>
	<form name="lock" action="admin.asp?op=admin_locked&amp;pass=1" method="post">
	<%=traducisys("ling_occ_180")%><br /><br />
    <%
    Set dRs = objConn.Execute("SELECT SYSTEM_STATUS_MSG FROM TB_REGEDIT_TAB") 
	msg = dRs("SYSTEM_STATUS_MSG")
	Set dRs = Nothing
	%>
	<textarea style="width:100%; height:400px;" class="editor" name="testo" cols="40" rows="10"><%=msg%></textarea>
	<%
	Call creaComponente("editor","full")
	%>
	<br />
	<input type="hidden" name="control" value="ok" />
	<br /><br />
	<input type="submit" class="buttonclass" value="<%=traducisys("ling_111")%>" />
	</form>
	<%
	End if
  CASE 1: 
   if request.Form("control") = "ok" then
    Set objFSO = Server.CreateObject("Scripting.FileSystemObject")
	if request.Form("testo") <>  "" then
		objConn.Execute("UPDATE TB_REGEDIT_TAB Set SYSTEM_STATUS = 1, SYSTEM_STATUS_MSG = '"&testSQLinj(request.Form("testo"))&"'") 
		response.Redirect("admin.asp?op=admin_locked")
    else
     response.Redirect "admin.asp?op=admin_locked"
    end if
   else
    response.Redirect "admin.asp?op=admin_locked"
   end if
  CASE 2:
	objConn.Execute("UPDATE TB_REGEDIT_TAB Set SYSTEM_STATUS = 0") 
	response.Redirect "admin.asp?op=admin_locked"
  End select
  %>
</div>